
1 Introduction
The following 3 exercises are part of the ALPHA work shop. They provide a brief introduction to
programming the Digilent BASYS FPGA boards using both a schematic design approach and a procedural
programming language called Verilog.
Though these exercises were based on the BASYS board – which is no longer sold by Digilent – and of
which I have a large number in my lab – the exercises will still work with newer boards. You will need to
adjust the pin assignments and account for the different system clock speed. I have listed this
information at the end of this handout for both the BASYS2 board and the NEXYS3 board.

2 Exercise 1:
Create a full adder using a schematic circuit representation

2.1 Background
The full adder circuit is the basis of all addition
circuits. Its truth table is:

0

1

1

Figure 1: Full Adder diagram (with inputs on the left and outputs on the right) and its truth table.

The truth table can be implemented using the following Boolean logic:
Sum_out = (A^B)^C, i.e., (A XOR B) XOR C
C_out = AB+AC+BC, i.e., (A AND B) OR (A AND C) OR ( B AND C)

Where, for brevity we replaced A_in with A and B_in with B etc. Its Boolean circuit representation is
shown below.

Figure 2: Full Adder Boolean Logic

We now implement this design on the FPGA on the BASYS board and assign buttons BTN0 through BTN2
for the inputs A_in, B_in and C_in; we then assign the output Sum_out to LED LD_0 and C_out to LD_1.

2.2 Step 1: Implement the Design in the ISE Webpack
Start the ISE Webpack. You find it under: Xilinx Design Tools \ ISE Design Suite 14.6 \ISE Design Tools \
64-bit Project Navigator.
Start a new project under: File / New Project and name it: (In this example the new project was named
“Full_Adder_Graphical.”)

Figure 3: New Project Window

Next you need to specify the particular FPGA for which this design is. Use the information shown below
for the particular FPGA Family, Device, Package and Speed on the BASYS board:

Figure 4: (FPGA) Device Properties

Hit “Finish” after the project summary and you should now see the empty ISE Project Navigator
environment as shown below.

Figure 5: Xilix ISE Webpack V 14.1

We now add a blank schematic file to our project on which we will design the project. Right click on the
upper right Design panel at the xc3s100e-4tq144 icon and select “New Source.”
The “New Source Wizard” window opens as shown below. Since we want to add a schematic file highlight the “Schematic” option in the “Source type” pull down menu and provide a file name. (In the
example shown it’s “FullAdderGraphical.”

Figure 6: Add new source: Source Wizard window

Hit “Next” and the “Finish” and you will be presented with a blank schematic page. (See below.) Select
the “Symbols” tab (see the figure below, arrow (1) ) so that you will have access to the Boolean logic
symbols to build the schematic.

(2)
(1)

Figure 7: Blank schematic page.

We will now build the circuit shown in Figure 2 by selecting the symbols from the list on the left and
then dropping them into the large empty panel on the right.
Begin by selecting the 2 input AND gates (see the figure below, arrow (2) ) by selecting “and2” from the
“Symbols” list and left click 3 times to place them at the appropriate place on the right panel, the empty
schematic page.
Repeat this for the two 2-input XOR gates, i.e., select “xor2” and the 3-input OR, i.e., “or3”.

Figure 8: Schematic with the basic gates

Before wiring the gates together its best to define and add the inputs and outputs to the diagram.
Select the “Add I/O Marker” tool from the tools palate (immediately to the left of the schematic panel,
see arrow in picture above) and place it at the inputs of each of the AND gates: place the cross hairs
directly at the input of the AND gate, mouse click left while holding the left mouse button down and
dragging it to the left. Finally release. (You may want to zoom in with the zoom tool.) See below.

Figure 9a: Schematic with I/O ports

Figure 9b: Same but ports renamed

For readability and to conform to our initial diagram, Figure 2, let’s rename the ports: right click on the
port label and select “Rename Port.” Type in the appropriate name in the window that pops up until
your schematic looks like the one shown in Figure 9b.
Finally, let’s wire the remaining connection. Select the wiring tool on the left, see the red arrow on the
figure below, (Add Wire tool) and complete your design. To beautify your design you may want to use
the “Select Tool,” the topmost tool, represented by an arrow icon, which will allow you to drag the
objects in the window.

Figure 10: Complete schematic with wires

2.3 Step 2: Assign the I/O Ports to Pins
Click on the “Design” tab on the left window panel to again get an overview of the entire project.
Right click on your topmost module (the one highlighted in the top left panel and preceded with the
three squares – see the blue arrow in Figure 10) and select “New Source.” This time select
“Implementation Constraints File” and name the user constraint (UC) file.

Figure 11: New Source Wizard with a user constrain (UC) file selected.

Hit “Next” and “Finish” and you will be shown a blank “UCF” file which in our case was named
“PinAssignment.ucf.”
The general syntax for pin assignments is:
NET "<port_name>" LOC=<pin_location>;

where <port_name> represents the name assigned to the port; for our design, this corresponds to A_in,
B_in, Sum_out etc. The pin location is obtained from the Digilent handout which is shown below for the
BASYS board. (Pin locations for BASYS2 and NEXYS3 boards are listed at the end of this handout.)

Figure 12: BASYS board with pin locations. NOTE: there is an error in the picture above: pins “P03” will not work; instead use
“P3.” Do this to all the other single digit pins where a zero has been inserted.

For example, switch SW0 has a pin location of P38, switch SW1 of P36 and LED LD0 of P15 etc.
Therefore, the correct notation for the UCF file will be:
NET "A_in" LOC = "P38";
NET "B_in" LOC = "P36";
NET "C_in" LOC = "P29";
NET "C_out" LOC = "P14";
NET "Sum_out" LOC = "P15";

Enter this into the blank UCF file. Make sure you type the port names exactly as listed in your schematic
design, including upper / lower case. We should now be done to compile our program.

2.4 Step 3: Generating the Program File
We are now ready to generate the bit file that will be uploaded to the FPGA by right clicking on the
“Generate Programming File” in the left panel under Processes. (See below.)

Figure 13: Compiling the code

If there are no mistakes you should see the “Process "Generate Programming File" completed
successfully” message at the bottom in the console window. The only thing left is now to transfer the bit
file or firmware to the FPGA.

2.5 Step 4: Transferring the Bit File to the FPGA using Digilent Adept
We will use the Digilent Adept software to transfer the bit file to the FPGA. Connect the BASYS board
with a USB cable to your computer and start the program Digilent / Adept / Adept.

8

Figure 14: Digilent Adept file transfer program

If the connection has been successful you should see the information displayed above with Adept
correctly identifying the board.
By default Xilinx will compile the code for storing it in the PROM of the BASYS board. (This is the method
we just used.)
To transfer the code, select the “Browse…” button under PROM and locate the bit file in your project
folder. Next select “Program” and the file will be transferred to the flash memory on the BASYS board.
(See the window below.)

Figure 15: Transferring the bit file into the PROM

When we use the PROM method, the bit file will remain in the flash memory until it is overwritten.
However, it is not read into the FPGA until the board has been reset by going through a power on/off
cycle.
To power the board on/off either disconnect the USB cable, use the power switch on the very left, SW8
or, preferably, hit the tiny reset button directly above the left most digit of the 7 segment LED display.

2.6 Step 5: Test It!
You can now test your design by selecting the three rightmost switches SW0 through SW2 and observing
the result of the addition in LD1 (MSB) and LD0 (LSB.)

2.7 4 Bit – Full Adder
To harness the full power of digital logic, simple designs like the previous full adder, are implemented
repeatedly and then linked together. For example, implementing the previous design 64 times and
connecting the correct inputs together could turn this design into a very fast and powerful 64-bit adder.
Even though the FPGA on the boards could easily handle the design, it’s the limitation of the input and
outputs on the BASYS boards that forces us to limit our design to a “modest” 4-bit adder which you will
build in the next sections. To do, so you will learn first how to “modularize” your design of the one-bit
full adder before reusing it in the 4-bit adder.
2.7.1 Creating Your Own Symbols / Modularize the Design
For convenience and readability, you may prefer to combine all the gates in one design into a single
module or function. (This is similar to creating a Sub-Vi in LabView.) You do so by selecting your design
of the full adder and then under Processes select: Processes / Design Utilities / Create Schematic Symbol
/ Run. Shown below is the symbol created for our full adder:

Figure 16: Newly created symbol from our schematic file

Once created and saved, it will be available through the symbols library where you now may use it
instead of the more detailed version showing the individual gates.

2.7.2 Building the 4-Bit Adder
After you have created the new schematic symbol, let’s re-use, i.e., “instantiate” it 4 times, to build the
4-bit adder shown below which will calculate the mathematical operation:
sum4 sum3 sum2 sum1 sum0 = a3 a2 a1 a0 + b3 b2 b1 b0
where the subscript 0 denotes the least significant bit, i.e., the LSB. For example the operation, 7 + 15 =
22 would be represented by:
Decimal Value

Binary Representation
7

10

0111

+15
=22

+ 1111
= 10110
Table 2.7.1.

Figure 17: 4-bit Full Adder circuit building on our previous Full Adder

In your design, you may permanently set c_in to 0 by connecting the corresponding input of the
instantiated module to 0, i.e., ground. (See the GND symbol used in Figure 17.) Use the table below to
assign the inputs and outputs to the correct pins.

Name
a0
a1
a2
a3
b0
b1
b2
b3
Sum0_out
Sum1_out
Sum2_out
Sum3_out
Sum4_out

Function
Input: LSB of a
Input: 1st bit of a
Input: 2nd bit of a
Input: MSB of a
Input: LSB of b
Input: 1st bit of b
Input: 2nd bit of b
Input: MSB bit of b
Result: LSB bit of the sum
Result: 1st bit of the sum
Result: 2nd bit of the sum
Result: 3rd bit of the sum
Carry-out and MSB of the sum

BASYS Control
Switch SW0
Switch SW1
Switch SW2
Switch SW3
Switch SW4
Switch SW5
Switch SW6
Switch SW7
LED LD0
LED LD1
LED LD2
LED LD3
LED LD4

Pin
P38
P36
P29
P24
P18
P12
P10
P6
P15
P14
P8
P7
P5

Table 2.7.2.

Note that the 4 right most switches represent term a, i.e., a3 a2 a1 a0, while the 4 left most switches
represent b, i.e., b3 b2 b1 b0. The LEDs should then directly correspond to the result that the addition of
the two terms produces. Implement your design and download it to the BASYS board. Test it by
comparing your results to ones shown in Table 2.7.1.
A complete version of this circuit can be found on the CD in the folder: Exercises\Full_Adder_Graphical3

2.8 Additional Notes:
2.8.1 Programming for RAM:
If you prefer, you can also compile the code directly into the RAM of the FPGA. In that case the FPGA
will use the bit file as soon as it has been transferred to the board. Of course, when the board is reset
the code is lost and must be sent again.
To compile for the RAM you must make the following changes:
1) Change the start-up clock from CCLK to
JTAG Clock by right-clicking in Processes
under “Generate Programming File” and
selecting “Process Properties.” You should
see the window in Figure 18 then select:
Category: Startup Options
2) Change Jumper JP3 on the BASYS board
from ROM to JTAG.

Figure 18: Changing the Startup Option to compile code
for RAM

3 Verilog
3.1.1 Verilog: Combinational Logic
Let’s repeat the 4-bit adder from the previous section. However, this time we will implement the entire
exercise using a hardware descriptive language, (HDL), such as Verilog.
You will find a version of this exercise in the Lab Manual, section 9.2. For the sake of time, however, we
will not implement it on the BASYS board, though you are of course, free to do so.
Shown below is the Verilog module of the full adder circuit, i.e., the Verilog implementation of Figure 2.
It consists of port declarations and the two combinational logic statements beginning with “assign.”
Since it has no physical pins assigned to it, the code will compile but a programming or bit file cannot
(yet) be generated.
module ALPHA_FullAdderModule(a, b, c, sum_out, c_out);
input a, b, c;
//the 3 input bits
output sum_out; //sum bit
output c_out;
//carry bit
assign sum_out = a^b^c;
assign c_out = (a&b)| (a&c) | (b&c);
endmodule

//a XOR b XOR c
//(A AND B) OR (A AND C) OR ( B AND C)

Shown below is the “topmost” module of our 4-bit adder. It works by “instantiating” (or calling) the full
adder module shown above 4 times as shown in the 4 lines highlighted in bold print.
module ALPHA_Full4bitAdderModule_SingleWireVersion(a0, a1, a2, a3, b0, b1, b2, b3, sum_out0, sum_out1, sum_out2,
sum_out3, sum_out4);
(* LOC = "P38" *)input a0;
(* LOC = "P36" *)input a1;
(* LOC = "P29" *)input a2;
(* LOC = "P24" *)input a3;
(* LOC = "P18" *)input b0;
(* LOC = "P12" *)input b1;
(* LOC = "P10" *)input b2;
(* LOC = "P6" *)input b3;
(* LOC = "P15" *)output sum_out0;
(* LOC = "P14" *)output sum_out1;
(* LOC = "P8" *)output sum_out2;
(* LOC = "P7" *)output sum_out3;
(* LOC = "P5" *)output sum_out4;
//Instantiate the ALPHA_FullAdderModule 4 times using the original module name followed by the new name
//NOTE: the ALPHA_FullAdderModule is not shown in this code segment, to save space
wire c_out0, c_out1, c_out2; //declaration of these (internal) wires is optional but highly recommnded
ALPHA_FullAdderModule ALPHA_FullAdderModuleInst0( a0, b0, 0,
sum_out0, c_out0);
ALPHA_FullAdderModule ALPHA_FullAdderModuleInst1( a1, b1, c_out0, sum_out1, c_out1);
ALPHA_FullAdderModule ALPHA_FullAdderModuleInst2( a2, b2, c_out1, sum_out2, c_out2);
ALPHA_FullAdderModule ALPHA_FullAdderModuleInst3( a3, b3, c_out2, sum_out3, sum_out4);
endmodule

Please note the following:
1)

The module assigns pins to the ports directly in the topmost module by using the (* LOC =
“PXX” *) syntax. Though we could have used a UCF file, identical to the previous graphical
exercise, this is another method for assigning pins. Both methods have their advantages and
disadvantages. We prefer the method shown above because we don’t have to deal with
another data file and it makes the entire process more transparent.
2) Wires c_out0, c_out1, etc., are internal wires. Therefore they do not need to be declared as
ports! It is optional to declare internal wires, i.e., if Verilog finds undeclared internal ports it
implicitly assigns them to a single bit internal wire. As long as you only have single bit wires, this
is ok. However, with multi-bit wires, i.e., vectors or buses, this leads to disasters and is usually
the leading cause for code not working! Therefore, get into the habit of declaring all internal
wires!
Since pins have been assigned to all the input and output ports, this code can now be compiled and a
bit-file can be generated from it.
Shown below is a more streamlined version of the module above using multi-bit wires, i.e., buses or
vectors:
module ALPHA_Full4bitAdderModule_VectorVersion(a, b, sum_out);
(* LOC = "P24 P29 P36 P38" *) input [3:0] a;
//4-bit vector / bus
(* LOC = "P6 P10 P12 P18" *) input [3:0] b;
//4-bit vector / bus
(* LOC = "P5 P7 P8 P14 P15" *) output [4:0] sum_out;
//5-bit vector / bus
//Instantiate the ALPHA_FullAdderModule 4 times using the original module name followed by the new name
//NOTE: the actual ALPHA_FullAdderModule is not shown in this code segment, to save space
wire c_out0, c_out1, c_out2; //declaration of these (internal) wires is optional but highly recommnded
ALPHA_FullAdderModule ALPHA_FullAdderModuleInst0( a[0], b[0], 0,
sum_out[0], c_out0);
ALPHA_FullAdderModule ALPHA_FullAdderModuleInst1( a[1], b[1], c_out0,
sum_out[1], c_out1);
ALPHA_FullAdderModule ALPHA_FullAdderModuleInst2( a[2], b[2], c_out1,
sum_out[2], c_out2);
ALPHA_FullAdderModule ALPHA_FullAdderModuleInst3( a[3], b[3], c_out2,
sum_out[3], sum_out[4]);
endmodule

Finally, look at the code below. It shows that Verilog is capable of basic (integer) math operations and,
hence, the entire 4-bit adder could also have been implemented in this much, much abbreviated,
simpler way by utilizing Verilog’s addition operator ‘+’:
module ALPHA_Full4bitAdderModule_Cheat(a, b, sum_out);
(* LOC = "P24 P29 P36 P38" *) input [3:0] a;
(* LOC = "P6 P10 P12 P18" *) input [3:0] b;
(* LOC = "P5 P7 P8 P14 P15" *)output [4:0] sum_out;
assign sum_out = a + b;
endmodule

4 Exercise 2: Radiation Counter
4.1 Introduction
In this exercise we will build a one-second reference timer that we then will use in the next section to
build a nuclear scaler that counts and displays the number of radiation events detected by a Geiger
Mueller (GM) tube. Finally, we will add the capability to detect coincidences between multiple GM
tubes.
The code for the complete GM counter consists of the following modules:
1) A reference clock, SecondsClock, which provides the timing signal every second.
2) A binary counter with an asynchronous reset, AsyncCounter, that will output the count obtained
when a reset signal is detected.
3) A display module, HexDisplayV1, that will display the binary counts on the seven segment
display in a decimal format.

Figure 19: Basic Radiation Counter Design

In this chapter you will only create and test the module for the reference clock. You will create the
other modules in the subsequent chapters.
This exercise closely resembles section 9.3.3 and 9.4. of the lab manual.

4.2 One Second Reference Clock
You will start the design by building the one-second reference timer:

Figure 20: One-Second Reference Timer Schematic

Use the 25 MHz signal from Pin 54 on the BASYS board and feed it into an n-bit counter. (See the flow
chart below.) Design the counter so that each time the counter reaches a value corresponding to an
elapsed time interval of 1 second, it resets itself to 0. Simultaneously, on each reset, it toggles an output
signal, toggle. (This signal is mainly used for debugging.) Assign this output signal to an LED and check

that your design works, i.e., every other second, the LED should be lit for exactly one second.
Additionally, create a second output called “reset” that stays HI for only one system clock cycle each
second.

Counter =
24999999 ?

TRUE

FALSE

Reset counter to 0
Toggle ‘toggle’
reset = 1

Increment counter
reset = 0

Name
sys_clk
toggle

reset

Function
Input: BASYS board 25 MHz clock
Toggles an LED on & off every other second. This output
is for debugging only since the “reset” output is difficult
to observe on a scope, due to its short duration.
Reset signal which stays on for only one system clock
cycle

BASYS Control
System Clock CLK1
LED: LD0

Pin
P54
P15

JD: D0

P58

Hints:
a) This design retains values and is dependent on previous values; therefore, it uses sequential
logic, i.e., registers.
b) It takes 25 bits to count to 24999999. Declare the counter register with 25 bits.
c) In your always block, use if-statements to determine if the counter has reached its 1 second
limit and needs to be reset or if it should be incremented further. On each reset, toggle
another, separate register which serves to control the LED.
d) Finally, create one more (output) register that is set HI only when the counter is reset.
Try to write the Verilog code that would accomplish this. You will find the answer in the next sections.

4.3 Step 1: Implement the Design in the ISE Webpack
Similar to the previous exercise, start a new project in the Xilinx Webpack using File / New Project. Enter
a new project name.

Before you continue, if you’re working on our lab computers, set the “Working Directory” to
“C:\Scratch\temp” as shown in the screen shot below. If you omit this step, the compiler will still work
though instead of taking a few seconds to generate the bit file, it takes minutes as it copies a large
number of files across the network!

Figure 21: Specifying the Project and Working Directory

Since you have already entered the specifications for the FPGA on the BASYS boards in the previous
exercise, you will not have to specify them again. Just hit “Next” till you get a new empty project similar
to Figure 5.
Add a blank Verilog file to our project. Right click on the upper right Design panel at the xc3s100e4tq144 icon and select “New Source.” (Again see Figure 5 for help.)
The “New Source Wizard” window opens as shown below. Since we want to use a Verilog file, select the
“Verilog Module” option in the “Source type” pull down menu and label and provide a file name. (In the
example shown it’s “SecondsClock.”) Hit “Next.”

Figure 22: The New Source Wizard: adding a Verilog file

The next window asks us to specify the input and output ports of our module. From the table above, we
can see that we have one input, “sys_clk” and two outputs “toggle” and “reset.” Indicate these choices
as shown in the screen shot below. (Be sure to select “output” for the “toggle” and “reset.”)

Figure 23: Specifying the I/O port names, directions and size.

Hit “Next” and “Finish” and you will return to the ISE Webpack editor, see below.

Figure 24: ISE Webpack with the empty module.

Note that the structure for your Verilog module has been created, specifying its inputs and outputs.
(See the picture above. If you discover a mistake, feel free to edit it) Also, if you prefer, you do not
have to use the wizard to enter the ports; instead, you can directly type the information in yourself.

At this point, the Verilog module is just an empty shell. We still need to fill in the specifics of our one
second reference clock algorithm. You may want to try this on your own and create the code for the
one-second timer; if you get stuck, detailed instructions are given in the following lines:
4.3.1.1 Detailed Instructions for the One-Second Timer
Start by creating a 25 bit counter called “mycounter.” Since the counter by definition is sequential logic,
after all it will depend on its previous state, we have to declare it as a 25 bit register:
reg [24:0] mycounter = 0;

Though it is optional, we also initialize it to 0.
Next we add the always@ structure that is activated at each positive edge of sys_clk cycle:
always@( posedge sys_clk) begin

end

Add these statements after the definition of the ports (after the semicolon) and before “endmodule.”
Finally we need to add the branching instruction inside the always block (using an if-else structure) that
determines if the counter is reset or incremented, toggle being toggled and reset being HI or LO.
reg [24:0] mycounter = 0;
always@(posedge sys_clk) begin
if( mycounter == 24_999_999 ) begin //underscores added for readability
mycounter <= 0;
//reset the counter
reset <= 1;
//set the 'reset' HI
toggle <= ~toggle;
//negate ‘toggle’ and assign it to itself
end
else begin
mycounter <= mycounter + 1;
//increment the counter
reset <= 0;
//set reset LO
end
end

The code is almost complete except for 2 issues:
First, when we entered our input and output ports, the compiler never inquired (or even gave us a
choice to specify) if they were of a sequential or combinational nature. (Note: while inputs ALWAYS are
of combinational nature, outputs can be either combinational or sequential, i.e., wires or a registers!)
Therefore, we need to correct that by preceding the port names with the keyword ‘reg.’ In the case
above, all our outputs are of sequential logic and hence must be declared as such. See the complete
code below with the changes highlighted:
module SecondsClock(
input sys_clk,
output reg toggle,
output reg reset );

reg [24:0] mycounter = 0;
always@(posedge sys_clk) begin
if( mycounter == 24_999_999 ) begin //underscores added for readability
mycounter <= 0;
//reset the counter
reset <= 1;
//set the 'reset' HI
toggle <= ~toggle;
end
else begin
mycounter <= mycounter + 1;
//increment the counter
reset <= 0;
//set reset LO
end
end
endmodule

We are not quite ready to “Generate the Program” file yet since we have not yet assigned the physical
pins to the I/O ports. We will do so in the next step.
(Warning: actually, if you made no mistakes, the program will generate the program file but in doing so,
it will assign random pins to the port which is worse than no pin assignment.)

4.4 Step 2: Define the I/O Port Pins
Similar to Step 2 in the previous design, we could employ a user constraint file to assign the pins to the
corresponding I/O ports. Instead we try a different approach and declare the pins directly within
Verilog. Though both methods work, the method shown here puts all the design information into one
single file.
The syntax for including the pin assignment inside of Verilog for a single port is:
(* LOC = "<XXX>" *) input YYY

where <XXX> represents the pin location and YYY the name of the port. For example:
(* LOC = "P54" *) input sys_clk,

You may also assign an entire bus to the corresponding pins as in the following example:
(* LOC = "P6 P10 P12 P18 P24 P29 P36 P38" *) input [7:0] x_in,

(Note: the rightmost pin, P38 in the above example, will correspond to the LSB and the leftmost value to
the MSB.)
Make those changes to your code so that the final version of your Verilog module will look something
like this:
module SecondsClock(
(* LOC = "P54" *) input sys_clk,
(* LOC = "P15" *) output reg toggle,
(* LOC = "P58" *) output reg reset
);
reg [24:0] mycounter = 0;

//LED LD0
//JD0

always@(posedge sys_clk) begin
if( mycounter == 24_999_999 ) begin //underscores added for readability
mycounter <= 0; //reset the counter
reset <= 1;
//set the 'reset' HI
toggle <= ~toggle;
end
else begin
mycounter <= mycounter + 1;//increment the counter
reset <= 0;
end
end

//set reset LO

endmodule

4.5 Step 3: Generating the Program File
Generating the bit file is identical to Step 3, section 2.4, of the previous exercise, i.e., right-click on
“Generate Program File” under Processes in the Design tab.

4.6 Step 4: Transferring the Bit File to the FPGA using Digilent Adept
Similarly, Step 4 is identical to the previous Step 4 except that the file name now corresponds to the
name of the module in this exercise. (In our case that is “Secondsclock.bit.”) You will find the bit file in
the folder that you specified as your working directory when you created your project. If you followed
the instructions here, that would be in the “C:\scratch\temp” folder.

4.7

Step 5: Test It!

After you have restarted or reset your board you are ready to test it. Does the right- most LED blink
every second? If so, continue to our next module.

5 Radiation Counter Exercise
In this section we will “recycle” the previous one-second timer module to build a “pretty good” radiation
survey instrument that displays the number of ionizing events measured in a one second time interval.

5.1 Radiation Detectors
The radiation detector, a Model RM 60, was manufactured by Aware Electronics. It utilizes a “cigar
shaped” (as opposed to a pancake shaped) Geiger Mueller tube to detect ionizing particles. Each time
the gas in the GM tube has been ionized by a particle passing through it, the unit generates a 50 μsec
negative going LVTTL pulse that is being sent to the BASYS board. It is these pulses that we want to
count when they trigger our radiation counter.
Up to four RM 60 units can be powered and interfaced to the BASYS board through the edge connectors
JA through JD as shown below.

Figure 25: RM 60 Radiation Monitor with Thorium Oxide gas mantel and BASYS board. The RM 60 has been connected to pin
1 of the BASYS board’s JA edge connector which corresponds to pin P81 on the FPGA.

The labels on top of the connector jacks indicate the pin number of the edge connector receiving the
pulse signal with the lowest number corresponding to BASYS board edge connector pin 1, for example
P81, and the highest number to pin 4, for example, pin P92. (Unlike the picture shown above, the
connectors are now labeled 0 through 3.) For this exercise, you will use only one detector. Connect it as
shown in the picture above and place it near a gas mantle. (Please do not unpack the gas mantles.)

These gas mantles contain Thorium Oxide. The Thorium itself emits a very low level of radiation as it
decays along its decay chain into lead. It is these decays that we will measure.
5.1.1 Step 1: Introduction
For this first step, you will start out with a very primitive radiation monitor which we will improve as we
go along.
Before we start, let’s look at the block diagram of the simplified circuit shown below. It consists of 3
modules:
1. BinaryCounter is 16-bit counter module that increments each time a radiation event is detected;
you will write the code for this module.
2. HexDisplayV1 is a module that displays the 16-bit counts in binary coded decimals on the
board’s 7-segment HEX display. The code for the module has already been written and will be
supplied to you.
3. The top-most module encapsulates, i.e., instantiates, the previous two modules; you will write
the code for this module.

Figure 26: Simple Counter with the top-most module explicitly shown by the dashed lines.

While students usually have no problem picking out the sub-modules, confusion often exists to the exact
nature of the top-most module, especially in regards to its ports. For example, should the bus
connecting the BinaryCounter and the HexDisplayV1 be declared or not? It’s for this reason that the
dashed outline was added to the schematic above. It explicitly denotes the borders of the top-module
and enhances the distinction between the external ports and the internal wires.
5.1.2 Step 2: Implementing the Top-Most module
Start with a completely new project. (If you forgot how, you may want to review section 4.3. of this
write-up.) Name your topmost module and carefully study the above diagram to determine which I/O
ports it requires. For example, should the 16 bit bus between the BinaryCounter’s counts and the
HexDisplayV1 value_in be listed? You should end up with a module that looks something like the one
shown below: (Of course, you may name your ports whatever you feel like as long as you understand
their purpose.)
module RadiationTopMod(
input sys_clk,
input rad_events,
//radiation events JA0
output [6:0] sevenSegLED_out,
//each bit corresponds to one of the 7 segments on the display
output [3:0] sevenSegPos_out//specifies which of the 4 displays is to be turned on (temporarily)

);
endmodule

5.1.3 Step 3: Add the HexDisplayV1 Module and Instantiate It
You have two options of adding additional modules to your project: you can either place them in
separate files and then them add them to your project (as explained below) or you can simply copy /
paste (or enter) the code of the new module into the same file as, for example, the topmost module. If
you choose the second method, it does not matter whether you enter the code before or after the topmost module as long as you have the topmost module declared!
For now, let’s add the file containing the display module to this project. Go to the menu bar and right
click on your topmost module and select Project \ Add Copy of Source and then select the HexDisplay1.v
file from the folder U:\pub_MXP\Verilog\HexDisplay to your current working Xilinx project directory.
(You can also find this file on the CD under \\Xilinx\Finished_Working\HexDisplay.) (Note: you must use
a copy of the file; linking to the (read-only) file in the pub_MXP directory will not work!)
The file should appear in the “Sources” window. Double click on it to open it. You will notice that it
contains multiple modules. The one that is of interest to us is the very first one, called HexDisplayV1.
(The other modules are support modules that will be instantiated by HexDisplayV1.) It has the following
inputs and outputs:
Name

I/O

sys_clk

Input

Bus
Size
(Bits)
1

value_in
BCD_enable

Input
Input

16
1

Display_Enable

Input

1

sevenSegLED_out Output 7
sevenSegPos_out

Output 4

Purpose

Pin(s)

25 MHz BASYS board system clock;
required for updating display.
Binary value to be displayed.
If this is set to 1, the display is in decimal
format; else, it is in hexadecimal.
If it is 1, the display is turned on or lit;
else it is turned off.
Output signal to the actual 7 segements
of each display.
Output signal to turn one of the 4
displays on.

P54

"P83 P17 P20 P21 P23 P16
P25"
"P26 P32 P33 P34"

Finally, instantiate the HexDisplayV1 module in your topmost module.
module RadiationTopMod(
input sys_clk,
input rad_events,
output [6:0] sevenSegLED_out,
output [3:0] sevenSegPos_out
);

//radiation events JA0
//each bit corresponds to one of the 7 segments on the display
//specifies which of the 4 displays is to be turned on (temporarily)

wire [15:0] counts_out;
HexDisplayV1 HexDisplayV1Inst(sys_clk, counts_out, 1, 1, sevenSegLED_out, sevenSegPos_out);
endmodule

5.1.4 Step 4: Add Binary Counter Module
Create a module that contains a 16-bit counter that is updated each time its signal input goes HI. (See
Figure 26.) You may add the code for this module directly after the topmost module.
module BinaryCounter(
input signal,
//count events to be measured, will be wired to rad_events
output reg [15:0] internal_counter
);
always@(posedge signal ) begin
internal_counter <= internal_counter + 1;
end
endmodule

5.1.5 Step 5: Continuous Radiation Counter
Instantiate the BinaryCounter module and add the pin numbers to the topmost module, as shown
below:
module RadiationTopMod(
(* LOC = "P54" *) input sys_clk,
(*CLOCK_DEDICATED_ROUTE = "FALSE", LOC = "P81" *) input rad_events,
//radiation events at JA_1 or connector labeled 0
(* LOC = "P83 P17 P20 P21 P23 P16 P25" *) output [6:0] sevenSegLED_out,
(* LOC = "P26 P32 P33 P34" *) output [3:0] sevenSegPos_out
);
wire [15:0] counts_out;
//important you must declare this bus!!!!! Why???????????
BinaryCounter BinaryCounterInst(rad_events, counts_out);
HexDisplayV1 HexDisplayV1Inst(sys_clk, counts_out, 1, 1, sevenSegLED_out, sevenSegPos_out);
endmodule

Before you generate the program file notice the CLOCK_DEDICATED_ROUTE = "FALSE", in the pin
assignment statement for the input rad_events. It was added to suppress an error message that the
Xilinx compiler generates when it encounters clock inputs that it deems “untrustworthy.” It is not clear
how it decides whether or not a source is suitable as a clock input but this statement demotes these
error messages to a warning.
Once you have transferred the bit file to the board, you should be able to observe the ionization events
as the counter keeps continuously incrementing. (Of course you must connect the radiation modules to
the BASYS board to observe anything.) From looking at the count rate, what do you estimate the count

rate per second to be? (Do not use detailed calculations; instead, just look at your counter and notice
which digits are updated about every second.)

5.2 Binary Counter with Asynchronous Reset
The previous version of the counter displayed a continuous count of ionization events. What we really
would like to observe is the number of counts in a given time period, for example, counts per second.
This can be achieved with two modifications / additions:
a) Add a reference time source that sends out a “trigger” signal every second.
b) Reset your counter at the beginning of each reference time period. Since the reset and
clock signals are independent of each other, this is considered an asynchronous reset.
(A synchronous reset would occur always at the same time as the clock signal.)
For a graphical representation on how to implement these ideas, see below.

Figure 27: Binary Counter with Asynchronous Reset

Implementing part a) is straightforward since you have already built the one-second reference timer in
chapter 4, exercise 2. You will instantiate it in your code and use its output signal to reset your radiation
events counter.
Part b) can be realized by modifying your already existing binary counter that keeps track of the
radiation events. First, add a reset input to the existing module.

Second, since the reset input is asynchronous it needs to be added to the list of clock signals in the
always @ block. See the example below for a positive edge reset signal.
always@(posedge signal or posedge reset ) begin

Third, since there are now two “clock” signals, signal and reset, an if-statement is required to discern
which source activated the always block. Specifically, if reset is HI, then the counter is reset to 0, else
the counter is incremented by one. Add these changes to your binary counter module. If you get stuck,
the complete code is shown below:
module AsyncCounter(
input signal,

//radiation events

input reset,
//reset signal for each second
output reg [15:0] internal_counter );
always@(posedge signal or posedge reset) begin
if( reset ) begin //"means:" if (reset is HI)
internal_counter <= 0;
end
else begin
internal_counter <= internal_counter + 1;
end
end
endmodule

Instantiate your new asynchronous binary counter and the one second clock module in the top-most
module and connect them as shown in the graphical representation above, Figure 27. Compile your
code, load it into the BASYS board and watch its operation. (If it doesn’t work correctly, for debugging,
you may want to connect the 1 second toggle clock output to LED LD0 on the BASYS board.) Do you
read a count rate similar to the one observed in the previous section?

5.3 Binary Counter with Asynchronous Reset and Storage Latch
The design still suffers from one flaw: though the counter now resets every second, it still displays a
continuously incrementing display. What we are really interested in is only the count value right before
it is reset. In other words, we need to store the count value in memory at the very same instant as the
counter is reset.
This latching action is accomplished by creating yet another 16-bit registry, called counts_out (see the
example below) that will be updated with the (internal) count value (internal_counter) right before it is
reset.
It is tempting to add the code for the latching action directly to the binary counter’s module always
block, specifically, to the section that handles the asynchronous reset, as shown below. This will not
work! You must create a separate always block (as shown further below) with only one clock source
that handles the store process.
module AsyncCounter(
input signal,
//radiation events
input reset,
//reset signal for each second
output reg [15:0] counts_out );
reg [15:0] internal_counter = 0;
always@(posedge signal or posedge reset) begin
if( reset ) begin //"means:" if (reset is HI)
counts_out <= internal_counter;
//latching action THIS WILL NOT WORK!
internal_counter <= 0;
end
else begin
internal_counter <= internal_counter + 1;

27

end
end
endmodule

As indicated the code above will not work properly and instead always will display a count value of 0.
The reason for it not working is very subtle. Can you figure out what it is? (Hint: read the lab manual
section 9.5.2. on Blocking vs. Non-blocking operators.) The working version of the code is shown below.
module AsyncCounter(
input signal,
//radiation events
input reset,
//reset signal for each second
output reg [15:0] counts_out );
reg [15:0] internal_counter = 0;
always@(posedge signal or posedge reset) begin
if( reset ) begin //"means:" if (reset is HI)
internal_counter <= 0;
end
else begin
internal_counter <= internal_counter + 1;
end
end
always@(posedge reset) begin
counts_out <= internal_counter;
end

//latching action

endmodule

Test your new circuit. You should now have a pretty good counter that can be used for other
applications. If you have problems you will find the complete code for this exercise on the CD in the
folder Exercises\Alpha_RadCounter\Alpha_RadCounter.zip

6 Radiation Counter with Adjustable Coincidence Window
6.1 Introduction
In this exercise, we modify the previous circuit to allow for coincidence measurement between two GM
detectors. We will instantiate the same modules as we are currently using in our Single Photon
Quantum Interference setup where we are measuring coincidences from up to 4 detectors. The results
of the measurements can be used to illustrate how the accidental coincidence rate from the two GM
tubes, Racc, depends on the width of the coincidence window τacc,, by:
Racc = 2 τacc R1 R2,
where R1 and R2 are the event rates of GM counters 1 and 2.
We will create a circuit that allows for a user-settable coincidence window width by using the switches
on the BASYS board. The complete schematic of the circuit is shown below.

Figure 28: Radiation counter with coincidence measurement between 2 GM tubes

The diagram above shows that two new modules were added to the design from the previous chapter,
namely CoincWindow and OneShotV2. While CoincWindow produces the user settable coincidence
window, OneShotV2 outputs a pulse of width, τacc, each time it has been triggered.
The main difference between this exercise and the previous ones where you entered the code is that
this time you will be given the complete code; instead you will learn how to analyze it using Xilinx’s built
in simulation capability and timing diagram program, ISIM. This is a vital tool for analyzing code and
debugging and the exercises in this chapter allow you to become familiar with it. For a screen shot of
the ISIM simulator, see Figure 29 below.

Figure 29: Timing Diagram from the Xilinx ISim Simulator

6.2 Xilinx ISim Timing Diagrams Using a Forced Clock Signal
The first part of any simulation is to create and add a source file called a “test bench” or “test fixture” to
a project. This file contains the port information for the unit under test (uut) as well as the timing and
clock stimulus information.
Once the test bench file has been created we have two options: we can either modify the file to include
the timing information in the form of a script language or we can use the interactive ISIM program to
create our signals. The advantage of the first option is that it allows for very complex signals. However,
the interactive approach using ISIM directly is perfectly suited for simple (forced) clock signals as in this
first exercise where we will analyze the CoincWindow module shown below:
//fout = fin/(2*(n+1)) where n is Switch setting
module CoincWindow( sys_clk, Switches, coinc_clk_out);
input sys_clk;
input [7:0] Switches;
output reg coinc_clk_out;
reg [7:0] int_counter = 0;
always@(posedge sys_clk) begin
if( int_counter == Switches) begin
int_counter <= 0;
coinc_clk_out <= ~coinc_clk_out; //toggle it
end
else
int_counter <= int_counter+1;
end
endmodule

You may want to spend a few seconds analyzing this code to see if you can predict its behavior.
Specifically, for a given input frequency, fo, (which is set by sys_clk), what is the width of the output
signal, coinc_clk_out, as a function of Switches?

If you think you understand the code, let’s start the ISIM simulator program following the steps outlined
below:
1. Start by creating a new project. Copy the RadCounterWCoincidenceV1a.v and HexDisplayV1.v
files from the folder U:\pub_MXP\Alpha Workshop FPGA
2013\CD\Exercises\Alpha_RadCounterCoicidence to your new project folder by using from the
Xilix menu Project >> Add Copy of Source. If everything is correct, RadiationCoinc1aTopMod
should show up as your top-module in the “Design Window.” (If not, set it by right clicking on it
and then selecting “Set as Top Module.”)
2. In the “Design Window,” right-click on the project and add “New Source;” select “Verilog Test
Fixture.” (See the screen shot below.) Name it and hit “next.”

Figure 30: Creating a Testbench File

3. In the next window, you are asked for the name of the source that you want to associate the
test bench with. Select the CoincWindow module. Hit “Next” and “Finish.”

4. In the Design Window, switch (1) from “Implementation” to “Simulation.” You should see the
newly added test bench. (2) (In the example shown, it has been called “CoincWindowTB.” (If
you're curious, you may want to glance at the text of the test bench file, shown in the right
window, though we will work more with the file in the next section.)
1

2

3

Figure 31: Simulation Mode

5. In the Process Window, right click on “Simulate Behavioral Model” (see figure above, arrow (3))
and select “Run.” The “Isim” window opens. (See Figure 29 – though no signals are displayed.)

6. In the ISim, from the menu, select Simulation >> Restart. All signals are cleared. See below.

Figure 32: Blank Isim

7. Now add a stimulus signal to the input signals that we are interested in, sys_clk and Switches.
Let’s start with sys_clk and add a forced clock stimulus: right click on it and select “Force Clock.”
You must specify the Leading & Trailing Edge (they must be opposite logic levels) and set the
Period to 1/25 MHz, i.e., use 40 ns. Hit “Apply” and “OK.” (See below.)

Figure 33: Clock Stimulus for the input sys_clk

8. We now add a constant value to the Switches input. Again right click on it, select “Force
Constant” and enter the values shown below. Note: we selected some arbitrary Switch setting
of 3.

Figure 34 Constant Stimulus for the input Switches

9. Finally, from the menu select Simulation >> Run. You may have to zoom out by pressing the
“Zoom to Full View” button to see the results since it covers the first 1000 nsec. (See arrow in
Figure 35 below.)

Figure 35

10. From the figure above, you can see that a switch setting of “3” creates an output that is 8 input
sys_clk cycles long. Now try different switch settings by changing the Force Constant to other
values, for example 4.
(Note: if you run the simulation again with new values, the simulation is appended to the
previously one. You can avoid that by selecting Simulation >> Restart. However, in that case all
input signal values are reset and you must re-enter all the values, such as the forced clock signal
again.)
11. The previous analysis can be greatly improved if we are able to observe signals that are internal
to the module. You can achieve that by first expanding and highlighting in the “Instant and
Process” window the “uut” (unit under test) icon. You should now observe on the window on
the right, “Simulation Objects for uut” all the internal signal. Drag and drop the

“int_counter[7:0]” into the timing diagram window to the right. Run the simulation again and
note now the display of the internal signal. If you don’t like its data representation, click on the
signal in the timing diagram window and select Radix >> Unsigned Decimal. You should see a
timing diagram similar to the one shown below.

Figure 36: Timing Diagram with Internal Signal

12. Finally, can you obtain the relationship between the given input frequency, fo, (which is set by
sys_clk), and the width of the output signal, coinc_clk_out, as a function of Switches?
In other words, can you verify from this (or the code) that: τacc = 2(n + 1)/fo where fo represents
the speed of the system clock (25 MHz) and n the decimal equivalent of the switch setting?
(Hint / Explanation: The CoincWindow module is very similar to our SecondsClock module. It
contains a counter which resets itself and toggles its output, coinc_clk_out, every time it reaches
a specific count value which is set by the toggle switches on the BASYS board. For example, with
a switch setting of 0000_0011, the internal counter will count at the speed of the system clock,
i.e., at 25 MHz, 0, 1, 2, 3, 0, 1, 2, 3, 0 etc.)
6.2.1 Additional Comments Regarding ISIM: Non Initialized Registers
While you are using the ISIM simulator you may change your Verilog code and see how it changes the
output. Let’s do that to illustrate one minor but important characteristic of the simulator, namely
uninitialized registers.
Go back to the ISE Webpack and open the source code. If it’s not already open, double click in the
project window on the source file. Find the following line of code in the CoincWindow module:
output reg coinc_clk_out = 0;

and change it to an uninitialized value:
output reg coinc_clk_out;

Save the code and return to ISIM and select Simulation >> Restart Simulation. Not surprisingly, all
signals are cleared and we will have to enter the Forced Clock for sys_clk and the Forced Constant for
Switches again.
However note one important change: the signal next to the coinc_clk_out is labeled “x” which means it
in indeterminate. Therefore, running the simulation will not produce any (useful) results since it needs
to know the initial conditions. (In the screenshot below, you can see that the output remains
indeterminate.)

There is really no good solution to this issue, except to go back to your code and make sure you assign
an initial value to all your registers!

6.3 ISim Simulator Using the Test Bench File
While the previous method works well for simulating repetitive signals, let’s look at a method that gives
us more flexibility over the signal stimulus. Instead of setting the stimulus in the graphical user
interface, it uses a script language in the test bench file to define the timing information, thereby,
allowing the creation of arbitrary timing signals. We will use this to create a clock stimulus for the
OneShotV2 module, shown below:
module OneShotV2( clk, asynctrigger_in, trig_out);
input clk;
input asynctrigger_in;
output trig_out; //lasts exactly one clock cycle
reg trig_set = 0;
reg trig_reset = 0;

always@(posedge asynctrigger_in or posedge trig_reset)
if( trig_reset)
trig_set <= 0;
else
trig_set <=1;
always@( posedge clk)
if( trig_set)
trig_reset <= 1;
else
trig_reset <=0;
assign trig_out = trig_reset;
endmodule

So let’s create another test bench file, this time for the OneShotV2 module. Follow steps 2, 3 and 4 in
the previous section but select the OneShotV2 module as the associated source.
Once the test bench file has been created, double click on it and open it. It should look similar to the
code shown below:
module MyOneShotTestBench;
// Inputs
reg clk;
reg asynctrigger_in;
// Outputs
wire trig_out;
// Instantiate the Unit Under Test (UUT)
OneShotV2 uut (
.clk(clk),
.asynctrigger_in(asynctrigger_in),
.trig_out(trig_out)
);
initial begin
// Initialize Inputs
clk = 0;
asynctrigger_in = 0;
// Wait 100 ns for global reset to finish
#100;
// Add stimulus here
end
endmodule

Let’s start by adding a constant clock signal to the test bench for our clk signal. We do so by employing a
powerful feature of the ISE Webpack, namely its Language Templates. This feature contains essential

37

code snippets that you can copy and paste into your code and then, with some minor modifications, use
without having to worry about the details of syntax.
Under the Edit Menu, select: Edit >> Language Templates. In the middle window “drill” down to Verilog
>> Simulation Constructs >> Clock Stimulus >> 50% Duty Cycle >> Using Always. (See the screen shot
below.) Copy the information in the rightmost window into the test bench directly before the
“endmodule” statement.

Note that the variables that you need to change so that they correspond to your project are usually
encased in braces such, as in the example above the parameter “<value>.” Modify the code snippet so
that it corresponds to our signals by setting the PERIOD to 40 and renaming the generic CLK name to clk,
as shown below:
……. Preceding lines omitted to save space…..
// Add stimulus here
end // Note: CLK must be defined as a reg when using this method
parameter PERIOD = 40;
always begin
clk = 1'b0;
//NOTE: Verilog is case sensitive so change “CLK” to “clk” here and the line below!
#(PERIOD/2) clk = 1'b1;
#(PERIOD/2);
end
endmodule

Though we have not yet defined our second input signal, the asynctrigger_in signal, let’s run the test
bench simulation to check that our clk signal works. (Make sure you save all your files.) In the Project
Explorer / Behavioral Window highlight your new test bench file. In the window below, right click on
“Simulate Behavioral Model” and select “Run.” ISIM will open and once you zoom out on the timing
window you should see the timing diagram shown below:

Now let’s add the stimulus for our asynchronous input trigger. Specifically, we would like to know:




How long is the period of the output trig_out signal?
Is it synchronized with the clk signal?
What happens if the one shot is triggered multiple times during one clk cycle, i.e., is it
retriggerable?

Close ISIM and go back to the test bench and add the trigger signals after the statement: “// Add
Stimulus here” in the form:
#xyz signal = value;

where xyz represents each subsequent time step (in nsec.)
For example, to produce a trigger signal that goes HI after 30 nsec and then LO 120 nsec later use the
following syntax:
#30 asynctrigger_in = 1;
#120 asynctrigger_in = 0;

Add these statements to the test bench directly after “// Add Stiumulus here.” as shown below:
// Wait 100 ns for global reset to finish

#100;
// Add stimulus here
#30 asynctrigger_in = 1;
#120 asynctrigger_in = 0;
end // Note: CLK must be defined as a reg when using this method
parameter PERIOD = 40;
always begin
clk = 1'b0;
#(PERIOD/2) clk = 1'b1;
#(PERIOD/2);
end
endmodule

Before you can re-run the ISIM simulator using the test bench method, you must first close it. Do so and
then in the ISE Webpack right click on “Simulate Behavioral Model” and select “Run.” You should see a
signal like the one shown below (once you zoom out.)

Figure 37

Though the trigger signal fulfills our specification, it is delayed by 100 nsec. This delay is caused by the
statement: “// Wait 100 ns for global reset to finish #100;” which was added (by default) to the test
bench when it was created. (This signal can be commented out if you prefer; for now, however leave it
in place.)
From the ISIM timing diagram we now observe that the one shot output signal is created in sync with
the first positive edge clk signal and that it lasts exactly for one clk period.

Now let’s check if the output is retriggerable, i.e, what happens if the input goes LO / HI during the HI
output? Will the output remain HI?
Add the following new delays to the trigger signal, highlighted in bold below.
………. PRECEDING CODE NOT SHOWN TO SAVE SPACE
// Add stimulus here
#30 asynctrigger_in = 1; //delay 30 nsec and then set HI
#120 asynctrigger_in = 0;
//delay 120 nsec then set LO etc.
#40 asynctrigger_in = 1;
#7 asynctrigger_in = 0;
#7 asynctrigger_in = 1;
#7 asynctrigger_in = 0;
end // Note: CLK must be defined as a reg when using this method
………CODE BELOW NOT SHOWN TO SAVE SPACE…

Run ISIM again and observe the following output signal. As can be seen from the screen shot below,
the one shot is non-retriggerable while it is HI. Go back to the Verilog code for this module and see if it
makes sense.

Figure 38 Input Retriggered

6.4 Coincidence Measurements
We should now understand how the entire coincidence counting program works and what the function
of each module is. (See the diagram below.)

The function of the OneShotV2 module is twofold: first it destroys any unwanted ringing or high
frequency noise from the detectors that could lead to false counts by suppressing any retriggering
during the coincidence window cycle. (Ringing and retriggering is a serious problem for FPGA counters
because they can operate at very high trigger rates and so it is very important to suppress such noise.)
Second, if we simply AND two signals with different coincidence window widths we might end up
counting multiple coincidences where there was only one; again, the one shot ensures that for every
event a signal with the same identical width is generated.
This is why we first send the detector signals to a one-shot module, called OneShotV2. This module
produces a non-retriggerable output signal with a duration set by τacc of the CoincWindow module
(which in turn can be controlled by the switches on the board.) Since the duration of this signal is
constant (at least as long as we do not change the switch settings) it can be ANDed and then sent to our
asynchronous counter.
(Note: the only difference to the Single Photon Counting module is that we also send the reset signal
from the one-second timer through the one-shot before it reaches the asynchronous counter. I am not
sure that is necessary though.)

6.5 Optional Exercise: Measure Accidental Count Rates
In this exercise we measure the accidental count rates from two GM tubes as a function of the
adjustable coincidence window width using the modules supplied in this and the previous chapter.
Get a strong radioactive source (gas mantels are too weak, Cs137 works fine) and measure the count
rate from two back-to-back GM counters with a radioactive source sandwiched between. First use the
program from chapter 5 to obtain the count rates, R1, and R2, for each of the two detectors.
Next use the program from this chapter and measure the coincidences as a function of the coincidence
window setting. (Connect the two GM tubes to the phone jacks labeled 0 and 3.) You should obtain the
following relationship:
Racc = 2 τacc R1 R2, with: τacc = 2(n + 1)/fo
where fo represents the speed of the system clock (25 MHz) and n is the decimal equivalent of the
switch setting on the BASYS board.
See below for a sample plot of the data:

Coincidence Counts vs. Coincidence Window

Figure 39 Sample Data (From Matthew Klein with Cs137 Source.)

6.6 A Final Comment and Warning about the ISIM Simulator
The ISIM simulator is a very powerful tool to analyze code. However, be aware that it produces an
output that is based on the (ideal) rules of digital logic. In reality, you can end up with code that works
perfectly fine in the simulator; however, that same code can produce unexpected behavior when
implemented which can be caused by glitches, noise and racing conditions in the real device!

7 Analog-to-Digital Converter using Pulse Width Modulation
implemented in Verilog
In this exercise we use Verilog to build a pulse width modulated (PWM) analog-to-digital converter
(ADC) to control the intensity of the LEDs on the board.

Figure 40: Diagram of a PWM control circuit. The Switch Control will be implemented using the FPGA.

The PWM switch control consist of an n-bit counter that continuously increments from 0 to its maximum
value, i.e., 2n-1 and then starts over. For this exercise we will use n = 8 bits, i.e., the counter will
continuously count from 0 to 255. The control input value “x_in” is set by the switches on the BASYS
board and its range is: 0 <= x_in <= 2n-1.
The switch control uses the following pseudo-code:
if ( counter < x_in )
PWM_out <= 1;
else
PWM_out <= 0;
counter <= counter+1;
We will implement this exercise using Verilog and we will use pin assignments inside the Verilog code
which will obviate the need for a user constraint file.

What follows are very detailed instructions for building this module. You may follow them or implement
the entire exercise on your own using only the information above and the following pin / port
assignments:
Port / Signal
clk_in
x_in (8 bits)
PWM_out

Direction
input
input
output

Pin(s)
P54
P6 P10 P12 P18 P24 P29 P36 P38
P15

Description
System clock (25 MHz)
Binary switches to set DAC value
LED LD0 whose intensity we will control
with the switches

If you encounter a problem read the instructions below.

7.1 Step 1: Implement the Design in the ISE Webpack
Similar to the previous exercise start a new project in the Xilinx Webpack using File / New Project. Enter
a new project name. Since you have already entered the specifications for the FPGA on the BASYS
boards in the previous exercise, you will not have to specify them again. Just hit “Next” until you get a
new empty project similar to Figure 5.
Add a blank Verilog file to our project. Right click on the upper right Design panel at the xc3s100e4tq144 icon and select “New Source.” (Again see Figure 5 for help.)
The “New Source Wizard” window opens as shown below. Since we want to use a Verilog file, select the
“Verilog Module” option in the “Source type” pull down menu and label and provide a file name. (In the
example shown it’s “AAPTSimplePWMVerilog.”) Hit “Next.”

Figure 41: The New Source Wizard: adding a Verilog file

The next window asks us to specify the input and output ports of our module. From the diagram above,
we can see that we have two inputs, “clk_in” and “x_in” and one output “PWM_out.” Furthermore,
“x_in” consists of multiple bits, i.e., it’s an 8 bit wide bus. Indicate these choices as shown in the screen
shot below. (Be sure to select “output” for the “PWM_out.”)

Figure 42: Specifying the I/O port names, directions and size.

Hit “Next” and “Finish” and you will return to the ISE Webpack editor, see below.

Figure 43: ISE Webpack with the empty module.

Note that the structure for your Verilog module has been created, specifying its inputs and outputs.
(See the picture above. If you discover a mistake, feel free to edit it) Also, if you prefer, you do not
have to use this wizard; instead, you can directly type the information in yourself.
At this point, the Verilog module is just an empty shell. We still need to fill in the specifics of the PWM
algorithm.
Let’s start by creating an 8 bit counter called “mycounter.” Since the counter by definition is sequential
logic, after all it will depend on its previous state, we have to declare it as an 8 bit register:
reg [7:0] mycounter;

Next we add the always@ structure that increments the counter at each positive edge of clk_in cycle:
always@( posedge clk_in) begin
mycounter <= mycounter + 1;
end

Add these statements after the definition of the ports (after the semicolon) and before “endmodule.”
Finally add the remaining lines of codes from the pseudo code above and check that you didn’t make a
mistake. (Hint: Is PWM_out a register or a wire? If it is a register, declare it accordingly, see below.)
The final and complete Verilog code reads: (with some of the potential pitfalls highlighted.)
module AAPTSimplePWMVerilog(
input clk_in,
input [7:0] x_in,
output reg PWM_out
);
reg [7:0] mycounter;
always@( posedge clk_in) begin
if ( mycounter < x_in )
PWM_out <= 1;
else
PWM_out <= 0;
mycounter <= mycounter + 1;
end
endmodule

We are not quite ready to “Generate the Program” file yet since we have not yet assigned the physical
pins to the I/O ports. We will do so in the next step.
(Warning: actually, if you made no mistakes, the program will generate the program file but in doing so,
it will assign random pins to the ports which is worse than no pin assignment.)

7.2 Step 2: Define the I/O Port Pins
Similar to Step 2 in the previous design, we could employ a user constraint file to assign the pins to the
corresponding I/O ports. Instead we try a different approach and declare the pins directly within
Verilog. Though both methods work, the method shown here puts all the design information into one
single file.
The syntax for including the pin assignment inside of Verilog for a single port is:
(* LOC = "<XXX>" *) input YYY

where <XXX> represents the pin location and YYY the name of the port. For example:
(* LOC = "P54" *) input clk_in,

You may also assign an entire bus to the corresponding pins as in the following example:

(* LOC = "P6 P10 P12 P18 P24 P29 P36 P38" *) input [7:0] x_in,

(Note: the rightmost pin, P38 in the above example, will correspond to the LSB and the leftmost pin to
the MSB.)
Make those changes to your code so that the final version of your Verilog module will look something
like this:
module AAPTSimplePWMVerilog(
(* LOC = "P54" *) input clk_in,
(* LOC = "P6 P10 P12 P18 P24 P29 P36 P38" *) input [7:0] x_in,
(* LOC = "P15" *)output reg PWM_out
);
reg [7:0] mycounter;
always@( posedge clk_in) begin
if ( mycounter < x_in )
PWM_out <= 1;
else
PWM_out <= 0;
mycounter <= mycounter + 1;
end
endmodule

7.3 Step 3: Generating the Program File
Generating the bit file is identical to Step 3, section 2.4, of the previous exercise, i.e., right-click on
“Generate Program File” under Processes in the Design tab.

7.4 Step 4: Transferring the Bit File to the FPGA using Digilent Adept
Similarly, Step 4 is identical to previous Step 4 except that the file name now corresponds to the name
of module in this exercise. (In our case that is “AAPTSimplePWMVerilog.”

7.5 Step 5: Test It!
After you have restarted or reset your board test it by operating the switches and observing the
brightness of the rightmost LED, LD0. With the rightmost switch being the LSB and the leftmost the MSB
can you gradually dim it?

7.6 Optional Exercise: Compare Simple PWM with Sigma Delta Algorithm
The Sigma-Delta algorithm provides a more sophisticated method to generate PWM and it can also be
used for voltage-to-frequency conversions. (For more details on the algorithm see the lab manual,
sections 10.2.) In this brief exercise we will use both algorithms and then display the resulting
waveforms simultaneously on the scope.
You may write your own Sigma-Delta PWM algorithm or you can copy the file PWM_Modules.v directly
from the U:\pub_MXP\Alpha Workshop FPGA 2013\CD\Exercises\Finished_Working\PWM to your
project.

Modify your existing top-module from section 7.5, the LED dimming exercise, and instantiate the
module SigmaDeltaPWM(clk_in, x_in, PWM_out) in the top-module.
“Wire” the Switches to the x_in input and sys_clk to clk_in of the newly instantiated module.
In the top-module, add two new output ports: use one for the simple PWM_out of your simple PWM
code and the other for the output of the Sigma-Delta PWM_out. Assign them to the pins of connectors
JD_1 and JD_4.
Compile your code and check the signals from the two algorithms on the scope. (To improve the quality
of the display, you should slow down the clock signal into the two algorithms. One way to achieve this is
to use the MSB of an 8-bit counter (which is driven by the sys_clk) as the new and much slower clock for
the two algorithms.)

8 Exercise 3: PWM Application: Simple 8-Bit Music Player
8.1 Background
In this exercise you will build a simple music player which reads a music file from a (preprogramed) flash
memory card and then converts it to an analog signal using the simple PWM algorithm developed in the
previous exercise.
In addition to creating a “real world application,” you will learn to instantiate (external) modules.
In addition to the previously written PWM module, this exercise will rely on the following modules:
1) AudioClock: The purpose of this module is to reduce the system clock from 25 MHz to an audio
frequency of 44.1 kHz by using a counter that sends out a HI signal every 567 system clock
cycles. (25 MHz/ 44.1 kHz = 567.) The 44.1 kHz frequency is the data rate at which the musical
data will be read from the flash memory board and sent to the PWM A2D converter.
1) AddressCounter: This generates the address for the byte of musical data to be read. It will
increment its counter each time its clk_in input goes HI, i.e., at the audio frequency. It does so
until it reaches its maximum value of 24'h1fffff (this notation means: 24 bits with hex value:
1fffff) at which point the counter is reset it to 0.
2) ReadOneByte: This module reads one byte of (musical) data from the flash memory at the
address specified at its address input. The byte is read each time its enable input goes HI. The
ReadOneByte module has already been written and will be provided to you.

Figure 44: General schematic of the full 8-bit music player. Note we will not use the Sigma-Delta PWM module; instead we
will use the simple PWM module from the previous exercise.

8.2 Step 1: Implement the Design in the ISE Webpack
Similar to the previous exercise, create a new empty project and add a new Verilog source file. (See
section 3.1.) Once the New Source Wizard opens specify the 6 I/O ports as shown in Figure 43 and 44.

Figure 45: The New Source Wizard with the port specifications.

Once the Verilog code for the top module has been generated, add the empty module declarations for
the AudioClock and the AddressCounter, i.e., just specify the modules’ I/O ports. Add this code directly
below the top module as shown below.
module AAPT_Audioplayer(
input sys_clk,
input Q,
output D,
output NCS,
output CLK_OUT,
output PWM_out
);

endmodule
//Generate a 44.1 kHz signal from a 25 MHz system clock
//Output every 567 system clock cycle (25E7/44.1E3 = 567) a an audio clock signal
module AudioClock(
input sys_clk,
output audio_clk
);

endmodule
// Increment the address counter for the flash memory until it
// reaches the maximum address, 24'h1fffff, and then reset it 0.
module AddressCounter(
input clk,
output [23:0] address_out
);
endmodule

Now implement the functionality of the AudioClock and the AddressCounter module. (See the
specifications in Section 4.1.) The newly added code is shown below in bold.
module AAPT_Audioplayer(
input sys_clk,
input Q,
output NCS,
output D,
output CLK_OUT,
output PWM_out
);

endmodule
//Generate a 44.1 kHz signal from a 25 MHz system clock
//Output every 567 system clock cycle (25E7/44.1E3 = 567) a an audio clock signal
module AudioClock(
input sys_clk,
output reg audio_clk //this output is a register
);
reg [9:0] counter;
always@(posedge sys_clk)
if( counter < 567 ) begin
counter <= counter + 1;
audio_clk <= 0;
end
else begin
counter <= 0;
audio_clk <= 1;
end
endmodule
// Increment the address counter for the flash memory until it
// reaches the maximum address, 24'h1fffff, and then reset it 0.
module AddressCounter(
input clk,
output reg [23:0] address_out
);
always@(posedge clk)
if( address_out < 24'h1fffff )
address_out <= address_out + 1;
else begin
address_out <= 0;
end
endmodule

Instantiate (or “call”) the AudioClock and the AddressCounter module from the top most module by
using the following syntax:
ExistingModule InstantiatedModule( a, b, c…);

The ExistingModule corresponds to the name of the module that you want instantiate; the
InstantiatedModule is the name of the newly instantiated module, which, really can be any name you
choose. Finally, within the parenthesis, list all the connections to the module’s I/O ports in the order
that they were listed in of the original (existing) module.
Shown below (in bold) are the instantiated AudioClock and AddressCounter modules and their
corresponding inputs.
module AAPT_Audioplayer(
input sys_clk,
input Q,
output NCS,
output D,
output CLK_OUT,
output PWM_out
);
AudioClock MyAudioClock( sys_clk, audio_clk );
AddressCounter MyAddressCounter( audio_clk, address ); //notice (wire) address
endmodule
//rest of code not shown to save space……………

To confirm that the instantiation worked, the newly instantiated modules are now shown in the process
window below the top most module. See the picture below. (You may have to save your files first to
update the Xilinx WebPack windows.)

Figure 46: ISE Webpack with instantiated modules.

To instantiate the ReadOneByte and the simple PWM copy the files FlashMemModules6.v and
AAPTSimplePWMVerilog.v (or whatever you named the file from the previous exercise) directly into your
project folder. These files can be found on the CD under U:\pub_MXP\Alpha Workshop FPGA
2013\CD\Exercises\AAPT_AudioPlayer1. Once they have been copied, right click on the top module and
select “Add source.” Add both of these files.
While it should be straight forward to instantiate the PWM module, instantiating the ReadOneByte
module can be daunting because you are not familiar with all its ports. (A description of its ports is
given in the table below.)
Direction Name
Size
Description
(Bits)
Input
clk_in
1
Serial interface clock for transmitting data; use the 25 MHz system
clock.
Input
enable
1
Hardware Trigger: at the positive edge, the new data is retrieved and
subsequently stored in registers “data.”
Input
address 24
Address where data is retrieved from in the flash memory.
Input
Q
1
Data read from memory is transmitted through this line.
Output
NCS
1
(Not) Chip-Select line.
Output
D
1
Data sent to the memory is transmitted through this line.
Output
clk_out 1
Clock signal from memory; required when receiving data.
Output
busy
1
While HI, flash memory is busy retrieving data; when it is LO again,
data has been stored in “data;” in our application, this output can be
ignored.
Output
data
8 or 16
Depending on version, either 8 or 16 bit register holding the data
value read from memory.

For the sake of brevity, its instantiation is shown below though you certainly should compare it with the
table above or the original file.
module AAPT_Audioplayer(
input sys_clk,
input Q,
output NCS,
output D,
output CLK_OUT,
output PWM_out
);
wire [23:0] address;
wire [7:0] data;
AudioClock MyAudioClock( sys_clk, audio_clk );
AddressCounter MyAddressCounter( audio_clk, address ); //notice (wire) address
ReadOneByte MyReadOneByte(sys_clk, audio_clk, address, Q, NCS, D, CLK_OUT, busy, data);
AAPTSimplePWMVerilog MyAAPTSimplePWMVerilog( sys_clk, data, PWM_out );
endmodule
//rest of code not shown to save space……………

Important: It’s worth noting here one very subtle point in the Verilog language that can cause a
tremendous amount of time to troubleshoot and it has to do with the implicit creation of wires. For
example, when instantiating module AudioClock, an output “audio_clk” was used which then was reused
at the inputs of the (instantiated) AddressCounter and the ReadOneByte module. Because “audio_clk”
was never explicitly declared in the code, Verilog implicitly assumes that it is a one bit wire. You can
explicitly declare it as a wire using the statement: wire audio_clk; but as stated earlier this is not
mandatory (though probably helpful.)
On the other hand, “data” and “address” both are buses consisting of 24, respectively 8 wires. If we had
not declared these buses explicitly as shown above, Verilog would have automatically assumed that they
were 1 bit wires. In doing so it would have truncated the remaining bits without any further warning.
(Well, if you read the warning messages in the console window carefully you might have noticed some
obscure message about this.) Nevertheless, this is a very common mistake so look out for it!

8.3 Step 2: Define the I/O Port Pins
Complete Verilog code using the following pin outs which will require you to plug the audio amplifier
into port A and the flash memory board into port C. (See photograph below.) The completed code for
the entire project is shown below:
module AAPT_Audioplayer(
(* LOC = "P54" *) input sys_clk,
(* LOC = "P76" *) input Q,
//flash mem module JC[3]
(* LOC = "P77" *) output NCS,
//flash mem module JC[1]
(* LOC = "P86" *) output D,
//flash mem module JC[2]
(* LOC = "P85" *) output CLK_OUT,
//flash mem module JC[4]
(* LOC = "P81" *) output PWM_out // audio out JA[1]
);
wire [23:0] address;
wire [7:0] data;
AudioClock MyAudioClock( sys_clk, audio_clk );
AddressCounter MyAddressCounter( audio_clk, address ); //notice (wire) address
ReadOneByte MyReadOneByte(sys_clk, audio_clk, address, Q, NCS, D, CLK_OUT, busy, data);
AAPTSimplePWMVerilog MyAAPTSimplePWMVerilog( sys_clk, data, PWM_out );
endmodule
//Generate a 44.1 kHz signal from a 25 MHz system clock
//Output every 567 system clock cycle (25E7/44.1E3 = 567) an audio clock signal
module AudioClock(
input sys_clk,
output reg audio_clk
);
reg [9:0] counter;
always@(posedge sys_clk)
if( counter < 567 ) begin
counter <= counter + 1;

audio_clk <= 0;
end
else begin
counter <= 0;
audio_clk <= 1;
end
endmodule
// Increment the address counter for the flash memory until it
// reaches the maximum address, 24'h1fffff, and then reset it 0.
module AddressCounter(
input clk,
output reg [23:0] address_out
);
always@(posedge clk)
if( address_out < 24'h1fffff )
address_out <= address_out + 1;
else begin
address_out <= 0;
end
endmodule

Figure 47: Screenshot of the final project

8.4 Additional Notes

The simple PWM algorithm will not work with the 2 bytes version because the conversion time
for this 16 bit A2D would be 25 MHz/216, i.e., 450 Hz resulting in a Nyquist frequency of 250 Hz.
In other words, all frequencies above this will be aliased to a frequency below 250 Hz resulting
in noise instead of music. For the 16 bit version to work with our board we need to use the
Sigma Delta version of the PMT. Try it.
The flash memory cards were programmed earlier by a program written in C that reads a WAV
file, converts it either to an 8 or 16 bit data stream which is then sent through RS232 to the
Verilog program that then writes it to the flash memory. Please contact me if you are interested
in the program. The musical selection was partially based on TA recommendations.

9 State Machines or Successive Approximation ADC Circuit
It is rare that you will need a complete microprocessor for a lab type FPGA project. Most of the times,
the projects can be implemented efficiently by using a modular approach and state machines. Since we
have already covered the modular approach and instantiation in previous chapters, this chapter is
dedicated to state machines. We will use such a state machine to implement a successive
approximation algorithm which we then use, together with a comparator circuit, to build a digital-toanalog converter.

9.1 Successive Approximation Analog-to-Digital Converter

Vanalog
comp_in

Successive
Approximation
Algorithm
(SAA)

Sigma/Delta
Digital-toAnalog
Converter

Vout

+

ADC_out

The diagram above shows the three main components of the successive approximation ADC. They are:
1. A successive approximation (SA) algorithm, which outputs a digital value, ADC_out.
2. A digital-to-analog converter which outputs an analog voltage, vout, which is proportional to
ADC_out.
3. A comparator circuit which sends a feedback signal, comp_in, back to the successive
approximation algorithm indicating whether the analog input voltage, vanalog, was larger or
smaller than vout.
Based on the value of comp_in, the successive approximation algorithm then adjust its output signal,
ADC_out, in ever decreasing steps until its analog voltage, vout, approaches that of the analog input
voltage, vanalog. Finally, when this state is reached, ADC_out corresponds to digital value of the analog
input voltage.
To reach the final condition the optimal way, the SA algorithm makes each voltage adjustment at ½ of
the value of the previous adjustment. In other words, while the SA algorithm starts out at half of its
maximum output signal, subsequent fine tunings are done at ¼, 1/8, 1/16, etc. of the maximum value.

58

This concept is straightforward to implement digitally since digital notation is based on the power of 2.
In other words, shifting bits within a word one position to the left (or right) corresponds to a
multiplication (or division) by 2. For example, binary 1000, 0100, 0010, 0001 corresponds in decimal to:
8, 4, 2, 1.
From this it follows that if we want to adjust our output signal in ever decreasing steps that are ½ of the
value of the previous adjustment, we need to start with the MSB and then work our way to the right
down to the LSB and at each at step compare and adjust our output voltage to the analog input voltage.
This idea is expressed in the flow chart below showing one complete cycle of the SA algorithm.
Start: Turn on the MSB
of the ADC_out signal.

Is vout exceeding the
analog input voltage, i.e.,
is the comparator
output, comp_in, HI or
LO?

vout > vanalog
comp_in is HI

Undo the previous action,
i.e., turns the bit off that was
previously turned on.

vout < vanalog
comp_in is LO

Has LSB been tested
yet?

no
Turn on the next lower bit.

yes

Done.

9.2 State Machine for the SA Algorithm

Figure 48: Schematic symbol of the SA algorithm module

The SA algorithm can be implemented in Verilog with a state machine with 4 distinct states. They are:
Name
INIT

Description
Initialize all the settings for the SA algorithm. Turn on the MSB for the ADC_out and
set the busy signal HI. Proceed to the COMP_ADJUST state.
COMP_ADJUST Check the comparator output, comp_in, and if it is HI then toggle the previously set bit
to LO. Also turn on the next lower bit. Finally check if the LSB has been previously set
and if so go to the DONE state otherwise continue with the ADJ_MASK state.
ADJ_MASK
Move the “masks” (see next section) to the next lower bit and go back to the
COMP_ADJUST state.
DONE
Set the busy signal LO; start the cycle over by going to the INIT state.
Figure 49: Table with states

9.2.1 Masking
One way of turning on, or toggling (off), specific bits within a word can be accomplished with a
technique called “masking.” Specifically, we will be using an 8-bit register masks that have been
initialized to: 1000_0000 or 0100_0000. For the lack of a better name, we will call them mask_now and
mask_next as shown in the following Verilog statement:
reg [7:0] mask_now = 8'b1000_0000;
reg [7:0] mask_next = 8'b0100_0000;

A specific bit within a word can then be turned on or off (without affecting any of the other bits) by:
a) shifting the 1 in the mask to the appropriate position by using the Verilog shift operator “>>”
and then:
b) OR-ing (to turn ON) or XOR-ing (to turn off) the mask and the word.
Study the Verilog example below:
reg [7:0] mask0 = 8'b1000_0000 ;
reg [7:0] mask1 = 8'b1000_0000 >> 4; //shift the mask 4 bits right, fill left side with 0. Final value of mask1: 8'b0000_1000
reg [7:0] some_word = 8’b0000_1111;
wire [7:0] wire0, wire1;
assign wire0 = some_word | mask0; //bit wise OR operation resulting in wire0 being assigned 8’b1000_1111
assign wire1 = some_word^mask1; //bit wise XOR operation resulting in wire1being assigned 8’b0000_0111

From this we can see that the actions required in the COMP_ADJUST state can be implemented by the
following statements:
if( comp_in )
//check if vout > vanalog
ADC_out <= (ADC_out ^ mask_now) | mask_next; //toggle the previously set bit to LO and set the next lower HI
else
ADC_out <= ADC_out | mask_next; // set the next lower bit HI (no need to undo previously set bit
//
since vout < vanalog)

Finally, shifting the masks one bit right during each step, as is required in the state ADJ_MASK can be
done with the right shift operator as shown here:
mask_now <= mask_now >> 1;
mask_next <= mask_next >> 1;

This should cover most of the details of the implementation so let’s start by programming our SA
algorithm using a state machine.

9.3 Verilog Implementation of the State Machine
Create a completely new project and a Verilog source file. As shown in figure 47, add the following ports
and registers to your SA algorithm module:
input sys_clk;
//required to “drive” the state machine
input comp_in;
//HI means vout too high. LO means ok.
output reg [7:0] ADC_out = 8'b1000_0000; //starting guess
output reg busy = 0;
//is HI during SA, and then LO when done
reg [7:0] mask_now = 8'b1000_0000;
reg [7:0] mask_next = 8'b0100_0000;

Instead of writing the entire state machine from scratch let’s copy it from the code snippets in the
Language Templates.
Under the Edit Menu, select: Edit >> Language Templates. In the middle window “drill” down to Verilog
>> Synthesis Constructs >> Coding Examples >> State Machines >> Moore >> Binary >>Safe >> 4 States.
(See the screen show below.) Copy all the information in the rightmost window into the test bench
directly before the “endmodule” statement. Look at the code and note that all signals encased in
greater than or smaller than operators, such as <clk>, must be replaced with names that match our
signals.

Figure 50: Language template for the state machine

Before we adjust the template to our needs notice the following 3 key points:
1) There are four states currently labeled <state1>, <state2>, <state3> and <state4>. You need to
replace these names (throughout the entire module) with names of your choosing. Select
descriptive names or the ones listed in table 49.
2) The main structure of the state machine consists of a case statement:
case (state).

At each posedge <clock>, one and only one of the four cases listed below, i.e., a one of the
states, is activated. (The <state1> case is listed below.)
<state1> : begin
if (<condition>)
state <= <next_state>;
else if (<condition>)
state <= <next_state>;
else
state <= <next_state>;
<outputs> <= <values>;
end

Which state is activated at posedge <clock> depends on what the register state has been set to
previously.
(For now, replace <clock> with the name of our input signal, sys_clk. Since we do not have a
“reset” input delete the 4 lines immediately following the always statement (shown below.)

//DELETE THESE LINES – THEY ARE NOT NEEDED
if (<reset>) begin
state <= <state1>;
<outputs> <= <initial_values>;
end

3) Now examine the individual cases in more detail and note that its functionality is set for two
different purposes, see below:
a. the decision which state the state machine will be at the next posedge <clock> and
b. the assignment of (various) values to (various) outputs.
<state1> : begin
if (<condition>)
state <= <next_state>;
else if (<condition>)
state <= <next_state>;
else
state <= <next_state>;
<outputs> <= <values>;
end

You will modify each case to suit the requirements for the particular state. While doing so, note
that there are situations where there is only one choice what the next state should be. In such a
situation, you may omit the if-statement and directly assign the next state to state.
Furthermore, in the template, only one assignment of a value to an output is shown. However,
you will find that multiple such statements will be needed in some of the states. In one case,
you will even need to make a conditional assignment.
See if you can implement the entire SA algorithm by modifying the language template. If you get stuck
you may “peek” at the completed example below which can also be found at U:\pub_MXP\Alpha
Workshop FPGA 2013\CD\Exercises\SuccessiveApprox
//////////////////////////////////////////////////////////////////////////////////
//8 bit succesive approximation module
//Note: only the very last approximation, i.e., the one after the busy goes low,
//represents the final approximation value. So latch it (on negedge) if you need it for the ADC value!
module SucApproxV2( sys_clk, comp_in, ADC_out, busy );
input sys_clk;
input comp_in;
//HI means too high, lower. LO means ok.
output reg [7:0] ADC_out = 8'b1000_0000; //current guess
output reg busy = 0;
reg [7:0] mask_now = 8'b1000_0000;
reg [7:0] mask_next = 8'b0100_0000;
parameter ADJ_MASK = 2'b00;
parameter INIT = 2'b01;
parameter COMP_ADJUST = 2'b10;
parameter DONE = 2'b11;

(* FSM_ENCODING="SEQUENTIAL", SAFE_IMPLEMENTATION="YES",
SAFE_RECOVERY_STATE="<recovery_state_value>" *) reg [1:0] state = INIT;
always@(posedge sys_clk)
(* PARALLEL_CASE, FULL_CASE *) case (state)
INIT : begin //initalize all parameters
state <= COMP_ADJUST;
mask_now <= 8'b1000_0000;
mask_next <= 8'b0100_0000;
ADC_out <= 8'b1000_0000;
busy <= 1;
end

COMP_ADJUST : begin
if (mask_now == 1)
state <= DONE;
else
state <= ADJ_MASK;

//adjust DAC output based on comp_in value; also set next lower bit

if( comp_in )
ADC_out <= (ADC_out ^ mask_now) | mask_next;
else
ADC_out <= ADC_out | mask_next;
end

ADJ_MASK : begin //adjust the masks which define the bits that are currently guessed
state <= COMP_ADJUST;
mask_now <= mask_now >> 1;
mask_next <= mask_next >> 1;
end

DONE : begin //send the conversion done message
state <= INIT;
busy <= 0;
end
default : begin // Fault Recovery
state <= INIT;
end
endcase
endmodule

9.4 Test Bench for SA Algorithm
To verify that your module works as expected, create a test bench. (See section 6.3.) Add a 40 nsec
Clock stimulus to sys_clk. Set the comp_in stimuli to 1010_0101, i.e., set it HI initially (no delay) then
wait one clock cycle each time to so set it LO, HI etc. see below.
initial begin
// Initialize Inputs
sys_clk = 0;
comp_in = 1;
//MSB
// Wait 100 ns for global reset to finish
#100;

//

// Add stimulus here
#80 comp_in = 0; //end of first guess, next higher bit
#80 comp_in = 1;
#80 comp_in = 0; // end of first four bits
#80 comp_in = 0;
#80 comp_in = 1;
#80 comp_in = 0;
#80 comp_in = 1;
end

From this you can see the final output of the ADC should be the inverse of comp_in, i.e., 0101_1010.
(Why?). Compare it with the timing diagram shown below.

Figure 51: Timing diagram for our SA algorithm

9.5 Complete SA Analog-to-Digital Converter
9.5.1 Circuit
We now put the SA algorithm to work and build the simple circuit shown below in Figure 52. It consists
of the following components:
To produce an arbitrary, adjustable analog voltge, vanalog, which we will measure with our ADC, we will
use a 10 turn 10k adjustable resistor labeled R1.

We will use a TLV2463 rail-to-rail comparator to compare our filtered output signal, vout, and vanalog. Its
output, comp_in, will be our feedback signal for our SA algorithm, which will be connected to pin JA_1.
We will use a sigma-delta algorithm to produce our DAC signal at pin JA_4 and we will filter it using the
380 Ohm and the 0.01 uF low pass filter.
Finally, both ground and 3.3V will be supplied by the BASYS board.

Figure 52: Successive Approximation Schematic (See below for a photograph of the circuit.)

9.5.2 Top Module
You may create your own top module to instantiate your SA algorithm. However, due to time
constraint, we already have created it for you and you may load it from: X:\pub_MXP\Alpha Workshop
FPGA 2013\CD\Exercises\SuccessiveApprox. Add the HexDisplayV1 and the PMW_Modules files to your
project.

Figure 53 SA ADC top module

In addition to the now familiar SucApproxV2, SigmaDeltaPWM and the HexDisplayV1 modules, the top
module consists of the following components: (see above)
IFD16 is a 16 bit register latch that stores the final vale of the SA algorithm (when the conversion is
done) so it can be displayed on the 7-segment display.

The ClockDividerV1 is a 12 bit counter whose most significant bit is used to slow down the clock for the
SA algorithm. Therefore, the SA algorithm runs at 1/4096 of the 25 MHz system clock speed. The
reason for this is that we need to integrate, i.e., filter the output of the SigmaDeltaPWM output over
multiple clock cycles to achieve a stable. i.e., smooth output. The 12 bit counter assures that the PWM
modules integrates over 4096 cycles, which frankly might be excessive, but is done to better illustrate
the working of the module.
After you have built the circuit and connected it to the BASYS board test it by adjusting the adjusting 10
turn pot. Does the seven segment display show the ADC value? You may also want to connect a scope
to Vout to better observe the workings the circuit.

Figure 54: Screenshot from the oscilloscope of vout

9.6 Final Comments
Though in this exercise we have limited ourselves to building a “simple” 8-bit ADC, the SA code could be
adjusted to accommodate an ADC of any other size. This can be accomplished by using the Verilog
“parameter” key word which acts similar to constant in C. A version of such a code (though not tested
yet) has been included in the file SucAproxV1 (directly above the top module) and is named

SucApproxMultibits. Though the parameter MAX_BITS is still set to 8, it could easily be changed in this
module as well the Sigma Delta PWM to accommodate a 16 or even 32 bit ADC and DAC. (Of course,
you would have to slow down the ClockDividerV1 algorithm accordingly and change the RC time
constants in the circuits.) Nevertheless, it illustrates the power and ease of HDL to expand on the
existing modules.

10 IP Cores
It may appear that we are constrained by HDL to low level tasks involving the manipulation of bits and
that we are forever stuck with simple integer arithmetic. It is the purpose of this last chapter to show
you that this is not case. Instead, we will examine and use an “IP core” to create a sine wave
synthesizer.

10.1 Introduction
So what is an IP Core? IP cores are to HDL what subroutines or libraries are to programming languages.
In other words, while they rely on the basic (digital) structure of HDL they allow the user to implement
sophisticated and complex tasks such as signal processing, filtering, data communication and
microprocessors that allows you turn your simple FPGA into a sophisticated instrument.
Xilinx does provide some free IP cores in the ISE Webpack. If you cannot find what you need, Xilinx and
other vendors will create one for your application, though probably not for free. In other words, the
business of IP cores is another important aspect of the entire FPGA concept.
To explore the types of free IP cores provided by the ISE Webpack, start a new project. In the process
window right click to “add a new source” and then select “IP CORE Generator and Architecture Wizard.”

The window above shows some of the IP Cores available for the Spartan 3E FPGA used on the BASYS
board. You may want to glance at all the options available but for now only want the IP Core with Trig
functions. Select the Digital Signal Processing >> Trig Functions >> DDS Compiler. After selecting “Next”
and “Finish,” the window shown below opens.

We need to specify the system clock rate (25.0 MHz) as well as the Spurious Free Dynamic Range at 44
dB to get an 8 bit sine wave. (Selecting the Spurious Free Dynamic Range determines the number of bits
calculated for the look-up table.)

Since we are only interested in the Sine output, we will disable the Cosine.

We disable the “Has Phase Out” since we’re not interested in that either.

For our last screen, we specify the output frequency of 0.001 MHz.

Two summary screens display various information before the program generates the actual IP Core.

This process may take a while but we ultimately should see the following options in the ISE Webpack
process window:

Figure 55: The newly generated IP Core in the ISE Webpack.

10.2 Implementing our Sinewave Synthesizer
Let’s create a top module that instantiates the Sinewave IP Core and then sends its output to our Sigma
Delta PWM (DAC) module. Add a new Verilog source to the project and add two ports to it: sys_clk and
PWM_out.

Figure 56 Sinewave Synthesizer Note: you must add 128 to sine(7:0) to change the output values into positive numbers.

Add a copy of the PWM_Modules.v modules to your project from U:\pub_MXP\Alpha Workshop FPGA
2013\CD\Exercises\Alpha_SineWave folder and instantiate the Sigma Delta module in the top module.
The only thing left, besides assigning pins, is to implement the IP Core. Again, the ISE Webppack
provides us generously with a template which we can find by clicking in the Process window on the
“View HDL Instantiation Template.” (See Figure 54 above.) It is shown below:
MySine your_instance_name (
.clk(clk), // input clk
.sine(sine) // output [7 : 0] sine
);

(You may not be familiar with the “dot” notation for inputs, as for example, “.clk(clk)” but by using this
notation you do not have to keep the order of the ports or even list all the ports during an instantiation.
For now, simply replace the signal names in braces with the names of your signal, such as for example:
.clk(sys_clk) )
Copy and paste the code into your top module. Link your internal signals, such as the sys_clk to the Sine
input, and the 8 bit wire “sine” output to the Sigma Delta PWM x_in input as shown in Figure 55.
Important: the output form the IP Core produces signed values; however, the input to the PWM module
expects only positive values, therefore, add the value 128 to the sine output as shown below:
SigmaDeltaPWM MySigmaDelta(sys_clk, x_in + 128, PWM_out);

(If you get stuck, the code of the top module is given at the end of this chapter.)
Assign pins for the sys_clk input and use JA_1 for PWM_out.
Compile the code and transfer it to your board. Connect a 39 kΩ resistor and a 0.01 uF capacitor to the
output to form a low pass filter. Connect it to the scope and observe your Sine function. See below for
a screen capture of the sine wave:

Figure 57: PWM Sine wave signal on the scope

One could turn this into a more sophisticated sine wave synthesizer by, for example, driving the MySine
IPcore with the programmable timer module from the adjustable coincidence width exercise, section
6.2.
module TopSine(sys_clk, PWM_out );
(* LOC = "P54" *) input sys_clk;

(* LOC = "P81" *) output PWM_out; //JA_1
wire [7:0] x_in;
SigmaDeltaPWM MySigmaDelta(sys_clk, x_in + 128, PWM_out);
MySine MyInstSine (
.clk(sys_clk), // input clk
.sine(x_in) // output [7 : 0] sine
);

endmodule

10.2.1 PWM Display
While figure 57 displays the LP filtered sine wave signal, you should indeed convince yourself that you
can create such a smooth signal by only using digital output signals, i.e., in our case 0V or 3.3V, and lowpass filtering. Therefore, you may find it instructional to observe the unfiltered output directly from
PWM_out. Make the following two changes to your program:
Since it is difficult to observe a “clean” 25 MHz signal from the PWM_out on the scope, slow down the
system clock by a factor of 256: insert an 8-bit counter into your program, clock it from sys_clk and use
the MSB of this counter to clock the SigmaDeltaPWM sys_clk and the MySine sys_clk inputs.
Second, add another port to the top-most module and connect the PWM_out to it. Assign pin P75,
JD_1, to it. Now observe the signal on the scope. Does it look like the simulation below? Does it make
sense?

10.3 Digital Clock Manager
The Digital clock manager IP Core allows you, among other tasks, to generate clock signals which are
significantly faster (and slower) than the input clock. For example, using the BASYS boards 25 MHz

system clock, we can generate clock signals up to 311 MHz! You may want to explore this feature on
your own by opening in the IP Core manager: FPGA Features and Design >> Clocking >> Spartan 3E & 3A
>> Single DCM_SP.
Specify the 25 MHz input frequency. The clock multiplier output is located at pin CLKFX, so select it in
the diagram as shown below.

Continue and select in the last screen, shown below, your desired output frequency:

Hit “Calculate” and then generate your IP Core and test it - you may have to placate the compiler with a
few ClOCK_DEDICATED_ROUTE = “FALSE” statements – but you should be able to observe your FPGA
now working at 300 MHz.

Closing Comments
10.4 Exercises
Some of the exercises presented here contain a short version of the lab assignments the students will do
in their 2 week Verilog lab. However, the chapters on graphical schematics, test benches and IP cores
were added to expose you to additional capabilities of the ISE Webpack are currently not part of the
course. For a more detailed discussion see our course lab manual which has been included on the CD.
Consult chapters 9 and 10 and appendix J and K. (An updated version is, as always, in progress.)

10.5 Board Choice
Though these exercises are based on the BASYS board (which is no longer sold by Digilent) they will work
on most other boards though you will need to adjust the pin assignments and account for the different
system clock speed.
For example, for the BASYS 2 boards use:
10.5.1 /* BASYS2 BOARD PIN ASSIGNMENT */
Use the “CP132” package (under Design Properties)
// Note: in multiple pin assignments, rightmost, as always, is LSB!
(* LOC = "B8" *) input sys_clk_in;
//50 MHz
(* LOC = "N3 E2 F3 G3 B4 K3 L3 P11" *) input [7:0] Switches;
(* LOC = "A7 M4 C11 G12" *) input [3:0] Buttons;
(* LOC = "M12 L13 P12 N11 N14 H12 L14" *) output [6:0] sevenSegLED_out;
(* LOC = "K14 M13 J12 F12" *) output [3:0] sevenSegPos_out;
(* LOC = "G1 P4 N4 N5 P6 P7 M11 M5" *)output [7:0] LEDs;
(* LOC = "B5 J3 A3 B2" *) input [3:0] JA_U;
(* LOC = "B7 C5 B6 C6" *) input [3:0] JB_U;
(* LOC = "C9 A10 B9 A9" *) input [3:0] JC_U;
(* LOC = "D12 C13 A13 C12" *) input [3:0] JD_U;

10.5.2 For Nexys3 use:
/ Nexys3: Family: Spartan 6; Device: XC6SLX16; Package: CSG324; Speed: -3
// Note: in multiple pin assignments, rightmost, as always, is LSB!
(* LOC = "V10" *) input sys_clk_in; //100 MHz
(* LOC = "T5 V8 U8 N8 M8 V9 T9 T10" *) input [7:0] Switches;
(* LOC = "C9 A8 D9 C4" *) input [3:0] Buttons;
(* LOC = "L14 N14 M14 U18 U17 T18 T17" *) output [6:0] sevenSegLED_out;
(* LOC = "P17 P18 N15 N16" *) output [3:0] sevenSegPos_out;
(* LOC = "T11 R11 N11 M11 V15 U15 V16 U16" *)output [7:0] LEDs;

(* LOC = "P11 N10 V12 T12" *) input [3:0] JA_U;
(* LOC = "V11 U11 N9 M10" *) input [3:0] JA_L;
(* LOC = "L3 L4 K1 K2" *) input [3:0] JB_U;

(* LOC = "K5 K3 J1 J3" *) input [3:0] JB_L;
(* LOC = "G3 K6 L7 H3" *) input [3:0] JC_U;
(* LOC = "F2 J6 J7 G1" *) input [3:0] JC_L;
(* LOC = "E11 F11 F10 G11" *) input [3:0] JD_U;
(* LOC = "E12 F12 C12 D12" *) input [3:0] JD_L;
//note:JX_U stands for the Upper set of pins, JX_L represents the lower row.

